

Task №2

STATIC PLANE THERMAL CONDUCTIVITY PROBLEM

KEYWORDS

1. Thermal conductivity problem (thermal analysis)
2. Static analysis
3. Plane problem

PROBLEM DESCRIPTION

Let us consider stationary thermal conductivity problem for two-dimensional domain in the shape of the letter “R”. The geometry of the domain in Oxy coordinate system is shown in Fig. 1. The sizes of the letter are defined by the following parameters: $HL=1.4$ m (height); $WL=0.8$ m (width); $WWL=0.2$ m (width of the legs); $HAL=0.7$ m (additional height size). The material of the domain is steel with the thermal conductivity $k= 46.7$ (W/(m·degree)).




Fig. 1. Scheme of the domain




Fig. 2. Numbering of keypoints and lines

The objective of the task is to perform thermal analysis and define the temperature and the heat flux vector distribution in the plane domain in the shape of the letter “R” under the following boundary conditions: the bottom edges of the domain are subjected to heating with prescribed temperature values and the top edge is subjected to heat transfer. Let us assume that at the left lower part of the boundary $0 \leq x \leq WWL$, $y = 0$, the temperature θ_h is equal to 90°C ; and at the right lower part of the

boundary $WL - WWL \leq x \leq WL$, $y = 0$, the temperature θ_h is equal to 180°C . The heat transfer is set at the upper part of the boundary $0 \leq x \leq WWL$, $y = HL$, with the value of the heat transfer coefficient 60 m^{-1} and the ambient temperature value 0°C . The rest of the boundary is thermally insulated.

INTRODUCTORY NOTES

It is necessary to note that the user should control the consistency of the system of units for the input values. Here the chosen system of units is SI, so the geometrical values are given in meters.

In order to build the domain in the shape of the letter “R” it is useful to note that its upper part is bounded by curved lines that can be defined as elliptical segments.

THEORETICAL BACKGROUND

Thermal conductivity equation with respect to the unknown function of temperature θ_h has the form:

$$\nabla \cdot (-\underline{\underline{K}} \cdot \nabla \theta_h) = 0, \quad (1)$$

where $\underline{\underline{K}}$ is the second rank tensor of thermal conductivities. The value $\underline{\underline{q}} = -\underline{\underline{K}} \cdot \nabla \theta_h$ is referred to as the heat flux vector. For isotropic material the tensor $\underline{\underline{K}}$ is diagonal, $K_{xx} = K_{yy} = K_{zz} = k$, and Eq. (1) can be rewritten as:

$$\nabla \cdot (-k \nabla \theta_h) = 0, \quad (2)$$

The condition of the convective heat transfer has the form:

$$\underline{\underline{n}} \cdot \underline{\underline{q}} = -h_f (\theta_{ext} - \theta_h), \quad (3)$$

where $\underline{\underline{n}}$ is the external unit normal to the boundary. For the example problem the heat transfer coefficient $h_f = 60 \text{ (m}^{-1}\text{)}$, and the ambient temperature is set to zero degrees: $\theta_{ext} = 0^\circ\text{C}$.

For thermally insulated boundaries the normal component of the heat flux vector is equal to zero:

$$\underline{\underline{n}} \cdot \underline{\underline{q}} = 0, \quad (4)$$

USING ANSYS TO SOLVE THE PROBLEM

For solving plane stationary thermal conductivity problem we will choose 6-node triangular thermal finite element PLANE35. This element has one degree of freedom in each node, temperature.

Fist we will build a solid model and after that we convert it to finite element model. To construct a solid model we will use a “bottom-top” approach, i.e. keypoints->lines->areas.

We will build two simply connected domains. The first domain will be bounded by the outer lines of the letter “R”, and the second domain will be bounded by the inner lines of the letter “R”. Then we will use Boolean operation of subtraction (**ASBA** command) to obtain the final domain in the shape of letter “R” with a hole.

In the following portion of the input file we define the title of the project (optional) with **/TITLE** command and in preprocessor **/PREP7** we define necessary parameters and their values.

```
/TITLE, Heat Flow in 2D region (R)
/PREP7
! Input data in SI system
! Geometrical sizes of the letter "R"
HL=1.4      ! height
WL=0.8  ! width
WWL=0.2 ! width of the walls of "R"
HAL=0.7 ! additional parameter
T_INP1=90 ! Temperature at the left lower boundary
T_INP2=180 ! Temperature at the right lower boundary
T_EXT=0    ! External temperature T_EXT=0
H_F=60     ! Convective heat transfer coefficient
```

Let us set material properties. For linear stationary thermal conductivity problem for isotropic homogeneous medium it is necessary to set only one value, that is the thermal conductivity coefficient $K=K_{XX}$. The value of K_{XX} can be defined in the first (and only) set of material properties by **MP** command (number “1” denotes the first set).

```
MP,KXX,1,46.7 ! Thermal conductivity coefficient KXX=46.7 (Steel)
```

Now we define the parameters of triangulation. We set one global size **DMESH** for all finite elements which is equal to the half of the leg width of the letter (**DMESH=WWL/2**). Another parameter **DMESH1**, which is four times smaller, will be used for mesh refinement.

```
! Parameters for finite element mesh
DMESH=WWL/2
DMESH1=DMESH/4
```

For the problem considered we have chosen to use triangular 6-node thermal finite element **PLANE35**. This will be the only type of elements that we will use in this analysis. **ET** command below will define the first (and only) type of finite elements as **PLANE35** (number “1” denotes element type).

```
ET,1,PLANE35 ! 6-node triangular thermal finite element PLANE35
```

From the scheme of the letter “R” it can be noted that its right upper part is bounded by the curved lines that can be defined as segments of

two ellipses. The center of the ellipses will coincide with the origin of Cartesian coordinate system $O'x'y'$ shifted relative to the initial Cartesian system Oxy by WWL to the right along the x-axis and by $(HL-HAL/2)$ up along the y-axis. The ratio of the semiaxes of the ellipse for the outer curved part of the letter “R” is equal to $(HAL/2)/(WL-WWL)$. It will be convenient to build curved lines in elliptical coordinate system so we define new elliptical coordinate system under number 11 on the base of cylindrical coordinate system (number 1) with the origin in the point $(WWL, HL-HAL/2)$ and the ellipticity parameter PEL .

PEL=(HAL/2)/(WL-WWL) ! Ellipticity parameter for the curved part of the figure

LOCAL,11,1,WWL,HL-HAL/2,,,,,PEL ! Elliptical coordinate system 11

The straight lines will be built in initial Cartesian coordinate system, so we change the number of active coordinate system to default value (0 – Cartesian system)

CSYS,0 ! Change active coordinate system to Cartesian

Let us construct the first domain bounded by external lines of the letter “R”. The domain is defined by nine keypoints:

! Define main keypoints for external boundary

K,1,0,0

K,2,WWL,0

K,3,WWL,HL-HAL-WWL

K,4,WL-WWL,0

K,5,WL,0

K,6,WWL,HL-HAL

K,7,WL,HL-HAL/2

K,8,WWL,HL

K,9,0,HL

Then we build lines by connecting keypoints. The curved lines can be built by switching to elliptical coordinate system 11 and using **L** command that constructs straight lines in active coordinate system (a straight line in elliptical coordinate system will be an elliptical segment). **AL** command builds the first area A1 by connecting lines.

! Define lines by end keypoints

L,1,2 \$ L,2,3 \$ L,3,4 \$ L,4,5 \$ L,5,6

CSYS,11 ! Change active coordinate system to coordinate system 11

L,6,7 \$ L,7,8

CSYS,0 ! Change active coordinate system to Cartesian

L,8,9 \$ L,9,1

AL,1,2,3,4,5,6,7,8,9 ! Define area 1 by lines

Note. Area A1 can be also constructed with the use of **A** command, that defines an area by connecting keypoints:

A,1,2,3,4,5,6,7,8,9 ! Define area 1 by connecting keypoints

If the lines between adjacent keypoints are already created, A command will use them. The missing lines will be generated "straight" in the active coordinate system.

When using A and AL command, it is necessary to know that the maximum number of arguments is 18 (keypoints) for A command and 10 (lines) for AL command. If an area to be built contains more than 18 keypoints or is bounded by more than 10 lines, than it can be convenient to create several areas instead of one, each containing appropriate amount of keypoint and lines. In this case the neighboring areas at common borders should share the same keypoints and lines.

The second domain bounded by internal lines of the letter "R" can be built in the same way.

!Define the hole - area 2

K,10,WWL,HL-HAL/2-(WL-2*WWL)*PEL

K,11,WL-WWL,HL-HAL/2

K,12,WWL,HL-HAL/2+(WL-2*WWL)*PEL

CSYS,11

L,10,11 \$ L,11,12

CSYS,0

L,12,10

AL,10,11,12

Using Boolean operation of subtraction we can construct area 3 by subtracting area 2 from area 1:

ASBA,1,2 ! Subtract area 2 from area 1

Thus, we have built a solid model of the letter "R" without boundary conditions. The resulting area 3 by default has all attributes with number "1", i.e. element type 1, material properties set 1 and so on. The element type and the material properties were defined previously by the commands **MP** and **ET**, respectively.

Let us move on from solid model to finite element model. First using **KESIZE** command we set global element edge length for elements around all keypoints. In order to refine the mesh around the hole, we set smaller element edge length around the keypoints at narrow places of the domain (see Fig. 2). **AMESH** command will built finite element mesh consisting of PLANE35 finite elements. **FINISH** command is used to finish the work with preprocessor (it can be used to exit any ANSYS processor)

! Define element sizes in the vicinity of the keypoints

KESIZE,ALL,DMESH

KESIZE,3,DMESH1 \$ KESIZE,6,DMESH1

KESIZE,10,DMESH1 \$ KESIZE,11,DMESH1 \$ KESIZE,12,DMESH1

```

AMESH,ALL ! Mesh all areas (area 3)
FINISH

```

The resulting finite element model is shown in Fig. 3 (use **EPLOT** command or Menu path: Plot->Elements).




Fig. 3. Finite element mesh

The next step is to set boundary conditions and solve the system of finite element equations. The actions can be done in the solver. **/SOLU** command enters the solver.

We will use **D** command to set the temperature values and **SF** command to set the convective heat transfer. Both commands act on nodes so it would be convenient to select the required nodes by **NSEL** command. After setting boundary conditions it is necessary to return to the selecting of all nodes of the model by issuing **NSEL,ALL**.

```

/SOLU
ANTYPE,STAT ! Static analysis
NSEL,S,LOC,Y,0 ! Select all nodes with coordinate Y=0
NSEL,R,LOC,X,0,WWL ! From selected set reselect all nodes with coordinate 0<=x<=wwl
D,ALL,TEMP,T_INP1 ! Define input temperature value for all selected nodes TEMP=T_INP1
NSEL,S,LOC,Y,0 ! Select all nodes with coordinate Y=0
NSEL,R,LOC,X,WL,WL-WWL ! From selected set reselect all nodes with coordinate wwl<=x<=wl-wwl
D,ALL,TEMP,T_INP2 ! Define input temperature value for all selected nodes TEMP=T_INP2
NSEL,S,LOC,Y,HL ! Select all nodes with coordinate Y=HL
NSEL,R,LOC,X,0,WWL ! From selected set reselect all nodes with coordinate 0<=x<=wwl
SF,ALL,CONV,H_F,T_EXT ! For all selected nodes define convective heat transfer

```

NSEL,ALL **! Select all nodes of the model**

Finally, we issue SOLVE command that forms finite element matrices and solves the system of linear algebraic equations of the finite element method. The results are saved to ANSYS database file under the name Heat_Letter_R.db.

SOLVE **! Решить СЛАУ МКЭ**
SAVE,Heat_Letter_R,db **! Сохранить базу ANSYS задачи под именем Heat_Letter_R.db**
FINISH

REVIEWING AND ANALYZING RESULTS

The finite element model of the problem with applied boundary conditions is shown in Fig. 4. (Menu path: Plot->Elements, for showing boundary conditions go to PltCtrls->Symbols->tick All applied BC, select showing convective heat transfer condition: Surface Load Symbols->Convective FilmCoef):




Рис. 4. Finite element model with boundary conditions

The results can be reviewed in the postprocessor. To enter general postprocessor, use command **/POST1**. Now we will look at several commands that govern graphical input. The same actions can be done via graphical user interface, GUI paths will be provided below.


```
/POST1  
! Commands for graphical output  
/SHOW,WIN32C
```

This command specifies the device and other parameters for graphics displays. Here graphics display WIN32c is specified. The default display is x11. GUI path: PlotCtrls->Device Options.

Next set of commands specify Window Options for graphical output.

```
/TRIAD,OFF ! Don't show the triad (origin of coordinates)  
/PLOPTS,INFO,2 ! Auto-legend format for caption  
/PLOPTS,LEG2,OFF  
/PLOPTS,LOGO,OFF ! ANSYS logo in text format  
/PLOPTS,FRAME,OFF ! Don't show frame  
/PLOPTS,TITLE,OFF ! Don't show title  
/PLOPTS,MINM,ON  
/PLOPTS,DATE,OFF ! Don't show date
```

In GUI Window options can be accessed through PlotCtrls->Window Ctrls->Window Options. Here one can choose the options for displaying the legend, Ansys logo, window frame, title, date and time, triad, etc.



The number of contour values is set in PlotCtrls->Style->Contours->Uniform contours->Number of contours, or by command:

/CONT,,14 ! Specifies the number of uniform contour values

In order to convert background from black to white one can issue commands

! Convert background from black to white

/RGB,INDEX,100,100,100,0

/RGB,INDEX,0,0,0,15

or use GUI path: PlotCtrls->Style->Colors->Reverse video

Now let us choose to display the boundary conditions for temperature and convective heat transfer at all pictures of results.

/PBC,TEMP,,1 ! display boundary conditions for temperature

/PSF,CONV,HCOEF,2 ! Display boundary conditions for thermal heat transfer with arrows

The following command plots the temperature distribution that is shown in Fig. 5.

PLNSOL,TEMP ! Plot temperature distribution

The same picture can be obtained in interactive mode by Menu path: General Postproc->Plot Results->Contour Plot->Nodal Solu->DOF Solution-> Nodal Temperature.




Fig. 5. Temperature distribution (with boundary conditions)

We can note from Fig.5 that minimal temperature value θ_h that approximately equals 51°C is observed at the upper boundary of the domain with the condition of convective heat transfer.

We can also obtain pictures of distribution of the secondary variables, such as the heat flux vector and its magnitude.

In order to plot the magnitude of the heat flux vector without showing boundary conditions, use the following commands:

```
/PBC,TEMP,,0 ! Don't show temperature boundary conditions  

/PSF,CONV,HCOEF,0 ! Don't show convective heat transfer boundary conditions  

PLNSOL,TF,SUM ! Plot magnitude of heat flux
```

In interactive mode, use Menu path: General Postproc->Plot Results->Contour Plot->Nodal Solu->Thermal flux->Thermal flux vector sum.

The picture of the magnitude of the heat flux vector distribution is shown in Fig. 6.




Fig. 6. Distribution of the heat flux vector magnitude

As it can be seen from Fig. 6, the heat flux vector has extrema at the corner points.

Fig. 7 shows the distribution of the heat flux vector (Menu path: General Postproc->Plot Results->Vector Plot->Predefined->Flux&gradient->Thermal flux TF).




Fig. 7. Distribution of the heat flux vector

The resulting input file that contains commands in ANSYS APDL is Th2LS_1.inp.

USING FLEXPDE TO SOLVE THE PROBLEM

Input file for solving the problem in FlexPDE is called St2LS_1.pde.